



# USEFUL FORMULA FIELDS

## Abstract

Provides examples of formula fields for various types of apps that you can use and modify for your own purposes.

## Examples of Advanced Formula Fields

---

This document contains custom formula samples for the following topics. For details about using the functions included in these samples, see [Formula Operators and Functions](#) on page 18.

- [Account Management](#)
- [Account Media Service Links](#)
- [Case Management](#)
- [Commission Calculations](#)
- [Contact Management](#)
- [Data Categorization](#)
- [Date and Date/Time Formulas](#)
- [Discounting](#)
- [Employee Services](#)
- [Expense Tracking](#)
- [Financial Calculations](#)
- [Image Links](#)
- [Integration Links](#)
- [Lead Management](#)
- [Metrics](#)
- [Opportunity Management](#)
- [Pricing](#)
- [Scoring Calculations](#)

## Account Management

---

### Account Rating

This formula evaluates Annual Revenue, Billing Country, and Type, and assigns a value of “Hot,” “Warm,” or “Cold.”

```
IF (AND (AnnualRevenue > 10000000,
CONTAINS (CASE (BillingCountry, "United States", "US", "America", "US",
"USA", "US", "NA"), "US")),
IF(ISPICKVAL(Type, "Manufacturing Partner"), "Hot",
IF(OR (ISPICKVAL (Type, "Channel Partner/Reseller"),
ISPICKVAL(Type, "Installation Partner")), "Warm", "Cold")),
"Cold")
```

In addition, you can reference this Account Rating formula field from the contact object using cross-object formulas.

```
Account.Account_Rating__c
```

## Account Region

This formula returns a text value of “North,” “South,” “East,” “West,” or “Central” based on the Billing State/Province of the account.

```
IF(ISBLANK(BillingState), "None",
IF(CONTAINS("AK:AZ:CA:HA:NV:NM:OR:UT:WA", BillingState), "West",
IF(CONTAINS("CO:ID:MT:KS:OK:TX:WY", BillingState), "Central",
IF(CONTAINS("CT:ME:MA:NH:NY:PA:RI:VT", BillingState), "East",
IF(CONTAINS("AL:AR:DC:DE:FL:GA:KY:LA:MD:MS:NC:NJ:SC:TN:VA:WV",
BillingState), "South",
IF(CONTAINS("IL:IN:IA:MI:MN:MO:NE:ND:OH:SD:WI", BillingState), "North",
"Other")))))
```

## Contract Aging

This formula calculates the number of days since a contract with an account was activated. If the contract Status is not “Activated,” this field is blank.

```
IF(ISPICKVAL(Contract_Status__c, "Activated"),
NOW() - Contract_Activated_Date__c, null)
```

## Account Media Service Links

---

### BBC™ News Search

This formula creates a link to a BBC news search site based on the Account Name.

```
HYPERLINK(
"http://newssearch.bbc.co.uk/cgi-bin/search/results.pl?scope=newsifs;tab=news;q=" & Name,
"BBC News")
```

### Bloomberg™ News Search

This formula creates a link to an account's ticker symbol on the Bloomberg website.

```
HYPERLINK(
"http://www.bloomberg.com/apps/quote?ticker=" & TickerSymbol,
"Bloomberg News")
```

### CNN™ News Search

This formula creates a link to a CNN news search site using the Account Name.

```
HYPERLINK(
"http://websearch.cnn.com/search/search?source=cnn&
invocationType=search%2Ftop&sites=web&query=" & Name,
"CNN News")
```

## MarketWatch™ Search

This formula creates a link to an account's ticker symbol on the Marketwatch.com website.

```
HYPERLINK (
"http://www.marketwatch.com/tools/quotes/quotes.asp?symb=" & TickerSymbol,
"Marketwatch")
```

## Google™ Search

This formula creates a link to a Google search site using the Account Name.

```
HYPERLINK (
"http://www.google.com/search?en&q=" & Name,
"Google")
```

## Google News Search

This formula creates a link to a Google news search site using the Account Name.

```
HYPERLINK (
"http://www.google.com/news?en&q=" & Name,
"Google News")
```

## Yahoo!™ Search

This formula creates a link to a Yahoo! search site using the Account Name.

```
HYPERLINK (
"http://search.yahoo.com/search?p=" & Name,
"Yahoo Search")
```

## Yahoo! News Search

This formula creates a link to a Yahoo! news search site using the Account Name.

```
HYPERLINK (
"http://news.search.yahoo.com/search/news?p=" & Name,
"Yahoo News")
```

# Case Management

---

## Autodial

This formula creates a linkable phone number field that automatically dials the phone number when clicked. In this example, replace "servername" and "call" with the name of your dialing tool and the command it uses to dial. The merge field, Id, inserts the identifier for the contact, lead, or account record. The first Phone merge field tells the dialing tool what number to call and the last Phone merge field uses the value of the Phone field as the linkable text the user clicks to dial.

```
HYPERLINK("http://servername/call?id="
& Id & "&phone=" & Phone, Phone)
```

## Case Categorization

This formula displays a text value of “RED,” “YELLOW,” or “GREEN,” depending on the value of a case age custom text field.

```
IF(DaysOpen__c > 20, "RED",
  IF(DaysOpen__c > 10, "YELLOW",
    "GREEN") )
```

## Case Data Completeness Tracking

This formula calculates the percentage of specific custom fields that contain data. The formula checks the values of two custom number fields: Problem Num and Severity Num. If the fields are empty, the formula returns the value “0.” The formula returns a value of “1” for each field that contains a value and multiplies this total by fifty to give you the percentage of fields that contain data.

```
(IF(ISBLANK(Problem_Num__c), 0, 1) + IF(ISBLANK(Severity_Num__c), 0, 1))
* 50
```

## Suggested Agent Prompts

This formula prompts an agent with cross-sell offers based on past purchases.

```
CASE(Product_Purch__c,
  "Printer", "Extra toner cartridges", "Camera", "Memory cards",
  "Special of the day")
```

## Suggested Offers

This formula suggests a product based on the support history for a computer reseller. When the Problem custom field matches a field, the formula field returns a suggestion.

```
CASE(Problem__c,
  "Memory", "Suggest new memory cards", "Hard Drive failure", "Suggest
  new hard drive with tape backup",
  "")
```

# Commission Calculations

---

## Commission Amounts for Opportunities

The following is a simple formula where commission is based on a flat 2% of the opportunity Amount.

```
IF(ISPICKVAL(StageName, "Closed Won"),
  ROUND(Amount * 0.02, 2), 0)
```

This example calculates the commission amount for any opportunity that has a “Closed Won” stage. The value of this field will be the amount times 0.02 for any closed/won opportunity. Open or lost opportunities will have a zero commission value.

### Commission Deal Size

This formula calculates a commission rate based on deal size, returning a 9% commission rate for deals over 100,000 and an 8% commission rate for smaller deals.

```
IF(Amount > 100000, 0.09, 0.08 )
```

### Commission Greater Than or Equal To

This formula assigns the "YES" value with a commission greater than or equal to one million. Note, this is a text formula field that uses a custom currency field called `Commission`.

```
IF(Commission__c >= 1000000, "YES", "NO")
```

### Commission Maximum

This formula determines what commission to log for an asset based on which is greater: the user's commission percentage of the price, the price times the discount percent stored for the account or 100 dollars. This example assumes you have two custom percent fields on users and assets.

```
MAX($User.Commission_Percent__c * Price,  
Price * Account_Discount__c, 100)
```

## Contact Management

---

### Contact's Account Discount Percent

This percent formula displays the account's `Discount Percent` field on the contacts page.

```
Account.Discount_Percent__c
```

### Contact's Account Name

This formula displays the standard `Account Name` field on the contacts page.

```
Account.Name
```

### Contact's Account Phone

This formula displays the standard `Account Phone` field on the contacts page.

```
Account.Phone
```

### Contact's Account Rating

Use this formula to display the `Account Rating` field on the contacts page.

```
CASE(Account.Rating, "Hot", "Hot", "Warm", "Warm", "Cold", "Cold", "Not  
Rated")
```

### Contact's Account Website

This formula displays the standard `Account Website` field on the contacts page.

```
Account.Website
```

If the account website URL is long, use the HYPERLINK function to display a label such as “Click Here” instead of the URL. For example:

```
IF(Account.Website="", "",
  IF(
    OR(LEFT(Account.Website, 7) = "http://", LEFT(Account.Website, 8) =
      "https://"),
    HYPERLINK( Account.Website , "Click Here" ),
    HYPERLINK( "http://" & Account.Website , "Click Here" )
  )
)
```

This formula also adds the necessary "http://" or "https://" before a URL if neither were included in the URL field.

## Contact's LinkedIn™ Profile

You can configure a link that appears on your contacts' profile page that sends you to their LinkedIn profile. To do so:

1. From Setup, click **Customize > Contacts > Buttons, Links and Actions**.
2. Click **New Button or Link**.
3. Enter a Label for this link, like LinkedInLink.
4. Enter this formula in the content box:

```
http://www.linkedin.com/search/fpsearch?type=people&keywords
={!Contact.FirstName}+{!Contact.LastName}
```

5. Click **Save**.

Remember to add this link to the Contact page layout in order for it to show up.

## Contact Identification Numbering

This formula displays the first five characters of a name and the last four characters of a social security number separated by a dash. Note that this example uses a text custom field called SSN.

```
TRIM(LEFT(LastName,
5)) & "-" & TRIM(RIGHT(SSN__c, 4))
```

## Contact Preferred Phone

This formula displays the contact's preferred contact method in a contact related list—work phone, home phone, or mobile phone—based on a selected option in a Preferred Phone custom picklist.

```
CASE(Preferred_Phone__c,
"Work", "w. " & Phone,
"Home", "h. " & HomePhone,
"Mobile", "m. " & MobilePhone,
"No Preferred Phone")
```

## Contact Priority

This formula assesses the importance of a contact based on the account rating and the contact's title. If the account rating is Hot or the title starts with Executive, then the priority is high (P1). If the account rating is Warm or the title starts with VP then the priority is medium (P2), and if the account rating is Cold then the priority is low (P3).

```
IF(OR(ISPICKVAL(Account.Rating, "Hot"), CONTAINS(Title, "Executive")),
"P1",
```

```
IF(OR(ISPICKVAL(Account.Rating, "Warm"), CONTAINS(Title, "VP")), "P2",
IF(ISPICKVAL(Account.Rating, "Cold"), "P3",
"P3")
)
)
```

### Contact Yahoo! ID

This formula displays a clickable Yahoo! Messenger icon indicating if the person is logged on to the service. Users can click the icon to launch a Yahoo! Messenger conversation with the person. This example uses a custom text field called `Yahoo Name` on contacts where you can store the contact's Yahoo! Messenger ID.

```
HYPERLINK("ymsgr:sendIM?"
& Yahoo_Name__c, IMAGE("http://opi.yahoo.com/online?u=" &
Yahoo_Name__c & "&m;=g&t;=0", "Yahoo"))
```

### Dynamic Address Formatting

This formula field displays a formatted mailing address for a contact in standard format, including spaces and line breaks where appropriate depending on the country.

```
CASE(ShippingCountry,
"USA",
ShippingStreet & BR() &
ShippingCity & ", " &
ShippingState & " " &
ShippingPostalCode & BR()
& ShippingCountry,
"France",
ShippingStreet & BR() &
ShippingPostalCode & " " &
ShippingCity & BR() &
ShippingCountry, "etc")
```

### Telephone Country Code

This formula determines the telephone country code of a contact based on the `Mailing Country` of the mailing address.

```
CASE(MailingCountry,
"USA", "1",
"Canada", "1",
"France", "33",
"UK", "44",
"Australia", "61",
"Japan", "81",
"?")
```

### Unformatted Phone Number

This formula removes the parentheses and dash characters from North American phone numbers. This is necessary for some auto-dialer software.

```
IF(Country_Code__c = "1", MID( Phone ,2, 3) & MID(Phone,7,3) &
MID(Phone,11,4), Phone)
```

## Data Categorization

---

### Deal Size Large and Small

This formula displays “Large Deal” for deals over one million dollars or “Small Deal” for deals under one million dollars.

```
IF(Sales_Price__c > 1000000,
  "Large Deal",
  "Small Deal")
```

### Deal Size Small

This formula displays “Small” if the price and quantity are less than one. This field is blank if the asset has a price or quantity greater than one.

```
IF(AND(Price<1,Quantity<1),"Small",
  null)
```

### Product Categorization

This formula checks the content of a custom text field named `Product_Type` and returns “Parts” for any product with the word “part” in it. Otherwise, it returns “Service.” Note that the values are case sensitive, so if a `Product_Type` field contains the text “Part” or “PART,” this formula returns “Services.”

```
IF(CONTAINS(Product_Type__c, "part"), "Parts", "Service")
```

## Discounting

---

### Maintenance and Services Discount

This formula field uses two custom currency fields: `Maintenance Amount` and `Services Amount`. It displays “Discounted” on an opportunity if its maintenance amount and services amount do not equal the opportunity `Amount` standard field value. Otherwise, it displays “Full Price.”

```
IF(Maintenance_Amount__c + Services_Amount__c <> Amount,
  "Discounted",
  "Full Price")
```

### Opportunity Discount Amount

This formula calculates the difference of the product `Amount` less the `Discount Amount`. Note that `Discount Amount` is a custom currency field.

```
Amount
- Discount_Amount__c
```



## Opportunity Discount Rounded

Use this formula to calculate the discounted amount of an opportunity rounded off to two digits. This example is a number formula field on opportunities that uses a custom percent field called Discount Percent.

```
ROUND(Amount-Amount*
Discount_Percent__c,2)
```

## Opportunity Discount with Approval

This formula adds a “Discount Approved” checkbox to an opportunity. It uses conditional logic to check the value of the approval flag before calculating the commission.

```
IF(Discount_Approved__c, ROUND(Amount - Amount * DiscountPercent__c,
2), Amount)
```

## Employee Services

---

### Bonus Calculation

This example determines an employee's bonus amount based on the smallest of two amounts: the employee's gross times bonus percent or an equally divided amount of the company's performance amount among all employees. It assumes you have custom number field for Number of Employees, a custom percent field for Bonus Percent, and currency custom fields for the employee's Gross and company's Performance.

```
MIN(Gross__c * Bonus_Percent__c,
Performance__c / Number_of_Employees__c)
```

### Employee 401K

This example formula determines which amount to provide in employee 401K matching based on a matching program of half of the employee's contribution or \$250, whichever is less. It assumes you have custom currency field for Contribution.

```
MIN(250, Contribution__c /2)
```

### Hours Worked Per Week

This formula uses a custom tab to enable time tracking of hours worked per day. It uses a formula field to sum the hours per week.

```
MonHours__c + TuesHours__c + WedsHours__c + ThursHours__c + FriHours__c
```

### Total Pay Amount

This formula determines total pay by calculating regular hours multiplied by a regular pay rate, plus overtime hours multiplied by an overtime pay rate.

```
Total Pay =
IF(Total_Hours__c <= 40, Total_Hours__c * Hourly_Rate__c,
40 * Hourly_Rate__c +
(Total_Hours__c - 40) * Overtime_Rate__c)
```

## Expense Tracking

---

### Expense Identifier

This formula displays the text “Expense-” followed by trip name and the expense number. This is a text formula field that uses an expense number custom field.

```
"Expense-"
& Trip_Name__c & "-" & ExpenseNum__c
```

### Mileage Calculation

This formula calculates mileage expenses for visiting a customer site at 35 cents a mile.

```
Miles_Driven__c * 0.35
```

## Financial Calculations

---

### Compound Interest

This formula calculates the interest, you will have after T years, compounded M times per year.

```
Principal__c * ( 1 + Rate__c / M ) ^ ( T * M )
```

### Compound Interest Continuous

This formula calculates the interest that will have accumulated after T years, if continuously compounded.

```
Principal__c * EXP(Rate__c * T)
```

### Consultant Cost

This formula calculates the number of consulting days times 1200 given that this formula field is a currency data type and consulting charges a rate of \$1200 per day. Note that Consulting Days is a custom field.

```
Consulting_Days__c
* 1200
```

### Gross Margin

This formula provides a simple calculation of gross margin. In this formula example, Total Sales and Cost of Goods Sold are custom currency fields.

```
Total_Sales__c - Cost_of_Goods_Sold__c
```

## Gross Margin Percent

This formula calculates the gross margin based on a margin percent.

```
Margin_percent__c * Items_Sold__c * Price_item__c
```

## Payment Due Indicator

This formula returns the date five days after the contract start date whenever Payment Due Date is blank. Payment Due Date is a custom date field.

```
(BLANKVALUE(Payment_Due_Date__c, StartDate  
+5)
```

## Payment Status

This formula determines if the payment due date is past and the payment status is "UNPAID." If so, it returns the text "PAYMENT OVERDUE" and if not, it leaves the field blank. This example uses a custom date field called Payment Due Date and a text custom field called Payment Status on contracts.

```
IF(  
AND(Payment_Due_Date__c < TODAY(),  
ISPICKVAL(Payment_Status__c, "UNPAID")),  
"PAYMENT OVERDUE",  
null )
```

## Image Links

---

### Yahoo! Instant Messenger™ Image

This formula displays an image that indicates whether a contact or user is currently logged in to Yahoo! Instant Messenger. Clicking the image launches the Yahoo! Instant Messenger window. This formula uses a custom text field called Yahoo Name to store the contact or user's Yahoo! ID.

```
IF(ISBLANK(Yahoo_Name__c), "", HYPERLINK("ymsgr:sendIM?" & Yahoo_Name__c,  
IMAGE("http://opi.yahoo.com/online?u=" & Yahoo_Name__c & "&m=g&t=0", "  
")))
```

### “Skype Me™” Auto Dialer Button

This formula displays an image that looks like a push button. Clicking the button automatically dials the specified phone number.

```
HYPERLINK("callto://" & "+1" & Phone,  
IMAGE("http://goodies.skype.com/graphics/skypeme_btn_small_blue.gif",  
"Click to Skype"))
```

### Flags for Case Priority

This formula displays a green, yellow, or red flag image to indicate case priority.

```
IMAGE(  
CASE( Priority,  
"Low", "/img/samples/flag_green.gif",  
"Medium", "/img/samples/flag_yellow.gif",
```

```
"High", "/img/samples/flag_red.gif",
"/s.gif"),
"Priority Flag")
```

## Color Squares for Case Age

This formula displays a 30 x 30 pixel image of a red, yellow, or green, depending on the value of a Case Age custom text field.

```
IF( Case_Age__c > 20,
IMAGE("/img/samples/color_red.gif", "red", 30, 30),
IF( Case_Age__c > 10,
IMAGE("/img/samples/color_yellow.gif", "yellow", 30, 30),
IMAGE("/img/samples/color_green.gif", "green", 30, 30),
))
```

## Traffic Lights for Status

This formula displays a green, yellow, or red traffic light images to indicate status, using a custom picklist field called Project Status. Use this formula in list views and reports to create a “Status Summary” dashboard view.

```
IMAGE (
CASE(Project_Status__c,
"Green", "/img/samples/light_green.gif",
"Yellow", "/img/samples/light_yellow.gif",
"Red", "/img/samples/light_red.gif",
"/s.gif"),
"status color")
```

## Stars for Ratings

This formula displays a set of one to five stars to indicate a rating or score.

```
IMAGE (
CASE(Rating__c,
"1", "/img/samples/stars_100.gif",
"2", "/img/samples/stars_200.gif",
"3", "/img/samples/stars_300.gif",
"4", "/img/samples/stars_400.gif",
"5", "/img/samples/stars_500.gif",
"/img/samples/stars_000.gif"),
"rating")
```

## Consumer Reports™—Style Colored Circles for Ratings

This formula displays a colored circle to indicate a rating on a scale of one to five, where solid red is one, half red is two, black outline is three, half black is four, and solid black is five.

```
IMAGE (
CASE(Rating__c,
"1", "/img/samples/rating1.gif",
"2", "/img/samples/rating2.gif",
"3", "/img/samples/rating3.gif",
"4", "/img/samples/rating4.gif",
"5", "/img/samples/rating5.gif",
"/s.gif"),
"rating")
```

## Horizontal Bars to Indicate Scoring

This formula displays a horizontal color bar (green on a white background) of a length that is proportional to a numeric score. In this example, the maximum length of the bar is 200 pixels.

```
IMAGE("/img/samples/color_green.gif", "green", 15, Industry_Score__c *
2) &
IMAGE("/s.gif", "white", 15,
200 - (Industry_Score__c * 2))
```

## Integration Links

---

### Application API Link

This formula creates a link to an application outside Salesforce, passing the parameters so that it can connect to Salesforce via the SOAP API and create the necessary event.

```
HYPERLINK ("https://www.myintegration.com?sId=" & GETSESSIONID() &
"?&rowID=" & Name & "action=CreateTask","Create a Meeting Request")
```

### Shipment Tracking Integration

This formula creates a link to FedEx, UPS, or DHL shipment tracking websites, depending on the value of a Shipping Method custom picklist field. Note that the parameters shown in this example for FedEx, UPS, and DHL websites are illustrative and do not represent the correct parameters for all situations.

```
CASE(Shipping_Method__c,
"Fedex",
HYPERLINK("http://www.fedex.com/Tracking?ascend_header=1&clienttype
=dotcom&cntry_code=us&language=english&tracknumbers= "&
tracking_id__c,"Track"),
"UPS",
HYPERLINK("http://wwwapps.ups.com/WebTracking/processInputRequest?HTMLVersion
=5.0&sort_by=status&loc=en_US&InquiryNumber1= "& tracking_id__c &
"&track.x=32&track.y=7", "Track") ,
"DHL",
HYPERLINK("http://track.dhl-usa.com/TrackByNbr.asp?ShipmentNumber=" &
tracking_id__c,"Track"), "")
```

### Skype™ Auto Dialer Integration

This formula creates a linkable phone number field that automatically dials the phone number via the Skype VOIP phone application. It requires installation of the Skype application (a third-party product not provided by salesforce.com) on your desktop.

```
HYPERLINK("callto://+" & Country_Code__c & Phone_Unformatted__c, Phone)
```

## Lead Management

---

### Lead Aging (for open leads)

This formula checks to see if a lead is open and if so, calculates the number of days it has been open by subtracting the date and time created from the current date and time. The result is the number of days open rounded to zero decimal places. If the lead is not open, this field is blank.

```
IF(ISPICKVAL(Status,
"Open"), ROUND(NOW()-CreatedDate, 0), null)
```

### Lead Data Completeness

This formula calculates the percent of certain lead fields that your sales personnel enter. The formula field checks the values of two custom number fields: `Phone` and `Email`. If the fields are empty, the formula returns the value "0." The formula returns a value of "1" for each field that contains a value and multiplies this total by fifty to give you the percentage of fields that contain data.

```
(IF(Phone = "", 0, 1) + IF>Email = "", 0, 1) ) * 50
```

### Lead Numbering

This formula returns a number value for the text value in the auto-number field `Lead Number`. This can be useful if you want to use the `Lead Number` field in a calculation, such as round-robin or other routing purposes. Note that auto-number fields are text fields and must be converted to a number for numeric calculations.

```
VALUE(Lead_Number__c)
```

### Round-Robin Assignment of Cases or Leads

The following formula example for leads assumes you have three lead queues and you want to assign an equal number of incoming leads to each queue. You can also assign cases using a similar formula.

```
MOD(VALUE(Lead_Number__c),
3)
```

This formula is for a custom formula field named `Round_Robin_ID` that assigns each lead a value of 0, 1, or 2. This formula uses a custom auto-number field called `Lead Number` that assigns each lead a unique number starting with 1. The `MOD` function divides the lead number by the number of lead queues available (three in this example) and returns a remainder of 0, 1, or 2. Use the value of this formula field in your lead assignment rules to assign lead records to different queues. For example:

- `Round_Robin_ID = 0` is assigned to Queue A
- `Round_Robin_ID = 1` is assigned to Queue B
- `Round_Robin_ID = 2` is assigned to Queue C

## Metrics

---

### Temperature Conversion

This formula converts Celsius degrees to Fahrenheit.

```
1.8 * degrees_celsius__c + 32
```

### Unit of Measure Conversion

This formula converts kilometers to miles.

```
Miles__c/.621371192
```

## Opportunity Management

---

### Expected Product Revenue

This formula calculates total revenue from multiple products, each with a different probability of closing.

```
ProductA_probability__c * ProductA_revenue__c + ProductB_probability__c  
* ProductB_revenue__c
```

### Maintenance Calculation

This formula calculates maintenance fees as 20% of license fees per year. Maintenance Years is a custom field on opportunities.

```
Amount * Maint_Years__c * 0.2
```

### Monthly Subscription-Based Calculated Amounts

This formula calculates an opportunity amount based on a monthly subscription rate multiplied by the subscription period.

```
Monthly_Amount__c * Subscription_Months__c
```

### Monthly Value

This formula divides total yearly value by 12 months.

```
Total_value__c / 12
```

### Opportunity Additional Costs

This formula calculates the sum of the product Amount, maintenance amount, and services fees. Note that Maint amount and Service Fees are custom currency fields.

```
Amount  
+ Maint_Amount__c + Services_Amount__c
```

## Opportunity Categorization

This formula uses conditional logic to populate an `Opportunity category` text field, based on the value of the `Amount` standard field. Opportunities with amounts less than \$1500 are “Category 1,” those between \$1500 and \$10000 are “Category 2,” and the rest are “Category 3.” This example uses nested IF statements.

```
IF(Amount < 1500, "Category 1", IF(Amount > 10000, "Category 3",
"Category 2"))
```

## Opportunity Data Completeness

This formula takes a group of fields and calculates what percent of them are being used by your personnel. This formula field checks five fields to see if they are blank. If so, a zero is counted for that field. A “1” is counted for any field that contains a value and this total is divided by five (the number of fields evaluated). Note that this formula requires you select the `Treat blank fields as blanks` option under `Blank Field Handling` while the `Advanced Formula` subtab is showing.

```
(IF(ISBLANK(Maint_Amount__c), 0, 1) +
IF(ISBLANK(Services_Amount__c), 0, 1) +
IF(ISBLANK(Discount_Percent__c), 0, 1) +
IF(ISBLANK(Amount), 0, 1) +
IF(ISBLANK(Timeline__c), 0, 1)) / 5
```

## Opportunity Expected License Revenue

This formula calculates expected revenue for licenses based on probability of closing.

```
Expected_rev_licenses__c * Probability
```

## Opportunity Revenue Text Display

This formula returns the expected revenue amount of an opportunity in text format without a dollar sign. For example, if the `Expected Revenue` of a campaign is “\$200,000,” this formula field displays “200000.”

```
TEXT(ExpectedRevenue)
```

## Opportunity Total Deal Size

This formula calculates the sum of maintenance and services amounts.

```
Amount + Maint_Amount__c + Services_Amount__c
```

## Opportunity Total Price Based on Units

This formula generates proposal pricing based on unit price and total volume.

```
Unit_price__c * Volume__c * 20
```

## Professional Services Calculation

This formula estimates professional service fees at an average loaded rate of \$1200 per day. `Consulting Days` is a custom field on opportunities.

```
Consulting_Days__c * 1200
```



## Stage-Based Sales Document Selection

This formula Identifies a relevant document in the Documents tab based on opportunity Stage. Use document IDs in the form of "001300000000;7AO."

```
CASE(StageName,
  "Prospecting", "Insert 1st Document ID",
  "Qualification", "Insert 2nd Document ID",
  "Needs Analysis", "Insert 3rd Document ID",
  "Value Proposition", ...
)
```

## Sales Coach

This formula creates a hyperlink that opens a stage-specific document stored in the Documents tab. It uses the previously defined custom formula field that identifies a document based on opportunity Stage. See [Stage-Based Sales Document Selection](#) on page 17.

```
HYPERLINK("/servlet/servlet.FileDownload?file=" & Relevant_Document__c,
  "View Document in New Window")
```

## Shipping Cost by Weight

This formula calculates postal charges based on weight.

```
package_weight__c * cost_lb__c
```

## Shipping Cost Percentage

This formula calculates shipping cost as a fraction of total amount.

```
Ship_cost__c / total_amount__c
```

## Tiered Commission Rates

This formula calculates the 2% commission amount of an opportunity that has a probability of 100%. All other opportunities will have a commission value of zero.

```
IF(Probability = 1,
  ROUND(Amount * 0.02, 2),
  0)
```

## Total Contract Value from Recurring and Non-Recurring Revenue

This formula calculates both recurring and non-recurring revenue streams over the lifetime of a contract.

```
Non_Recurring_Revenue__c + Contract_Length_Months__c *
Recurring_Revenue__c
```

## Pricing

---

### Total Amount

This formula calculates a total amount based on unit pricing and total units.

```
Unit_price__c * Total_units__c
```

### User Pricing

This formula calculates a price per user license.

```
Total_license_rev__c / Number_user_licenses__c
```

## Scoring Calculations

---

### Lead Scoring

This formula scores leads, providing a higher score for phone calls than website requests.

```
CASE(LeadSource, "Phone", 2, "Web", 1, 0)
```

Here's a formula that scores a lead based on his or her rating:

```
CASE(1, IF(ISPICKVAL(Rating, "Hot"), 1, 0), 3, IF(ISPICKVAL(Rating, "Warm"), 1, 0), 2, IF(ISPICKVAL(Rating, "Cold"), 1, 0), 1))
```

### Customer Success Scoring

This formula uses a simple scoring algorithm to rank customers a high score for positive survey results in Salesforce.

```
Survey_Question_1__c * 5 + Survey_Question_2__c * 2
```

## Formula Operators and Functions

---

Use the following operators and functions when building formulas. Click on the name of the operator or function below to view more details. All functions are available everywhere that you can include a formula such as formula fields, validation rules, approval processes, and workflow rules, unless otherwise specified.



**Note:** Extraneous spaces in the samples below are ignored.

## Math Operators

| Operator   | Description   |
|--|---|
| <b>+</b> (Add)                                     | Calculates the sum of two values.   |
| <b>-</b> (Subtract)                                | Calculates the difference of two values.  |
| <b>*</b> (Multiply)                                | Multiplies its values.  |
| <b>/</b> (Divide)                                  | Divides its values.   |
| <b>^</b> (Exponentiation)                          | Raises a number to a power of a specified number.   |
| <b>()</b> (Open Parenthesis and Close Parenthesis) | Specifies that the expressions within the open parenthesis and close parenthesis are evaluated first. All other expressions are evaluated using standard operator precedence. |

## Logical Operators

| Operator                             | Description  |
|--------------------------------------|--|
| <b>= and ==</b> (Equal)              | Evaluates if two values are equivalent.  |
| <b>&lt;&gt; and !=</b> (Not Equal)   | Evaluates if two values are not equivalent.  |
| <b>&lt;</b> (Less Than)              | Evaluates if a value is less than the value that follows this symbol.  |
| <b>&gt;</b> (Greater Than)           | Evaluates if a value is greater than the value that follows this symbol.   |
| <b>&lt;=</b> (Less Than or Equal)    | Evaluates if a value is less than or equal to the value that follows this symbol.  |
| <b>&gt;=</b> (Greater Than or Equal) | Evaluates if a value is greater than or equal to the value that follows this symbol.   |
| <b>&amp;&amp;</b> (AND)              | Evaluates if two values or expressions are both true. Use this operator as an alternative to the logical function AND.               |
| <b>  </b> (OR)                       | Evaluates if at least one of multiple values or expressions is true. Use this operator as an alternative to the logical function OR. |

## Text Operators



| Operator                   | Description                   |
|----------------------------|-------------------------------|
| <b>&amp;</b> (Concatenate) | Connects two or more strings. |

## Date and Time Functions

| Function             | Description   |
|----------------------|---|
| <b>DATE</b>          | Returns a date value from year, month, and day values you enter. Salesforce displays an error on the detail page if the value of the DATE function in a formula field is an invalid date, such as February 29 in a non-leap year. |
| <b>DATEVALUE</b>     | Returns a date value for a date/time or text expression.  |
| <b>DATETIMEVALUE</b> | Returns a year, month, day and GMT time value.  |
| <b>DAY</b>           | Returns a day of the month in the form of a number between 1 and 31.  |

| Function     | Description   |
|--------------|---|
| <b>MONTH</b> | Returns the month, a number between 1 (January) and 12 (December) in number format of a given date. |
| <b>NOW</b>   | Returns a date/time representing the current moment.  |
| <b>TODAY</b> | Returns the current date as a date data type.   |
| <b>YEAR</b>  | Returns the four-digit year in number format of a given date.                                       |

## Informational Functions

| Function          | Description  |
|-------------------|--|
| <b>BLANKVALUE</b> | Determines if an expression has a value and returns a substitute expression if it does not. If the expression has a value, returns the value of the expression.  |
| <b>ISBLANK</b>    | Determines if an expression has a value and returns TRUE if it does not. If it contains a value, this function returns FALSE.  |
| <b>ISNULL</b>     | Determines if an expression is null (blank) and returns TRUE if it is. If it contains a value, this function returns FALSE.<br><br> <b>Important:</b> Use ISBLANK instead of ISNULL in new formulas. ISBLANK has the same functionality as ISNULL, but also supports text fields. Salesforce will continue to support ISNULL, so you do not need to change any existing formulas.   |
| <b>NULLVALUE</b>  | Determines if an expression is null (blank) and returns a substitute expression if it is. If the expression is not blank, returns the value of the expression.<br><br> <b>Important:</b> Use BLANKVALUE instead of NULLVALUE in new formulas. BLANKVALUE has the same functionality as NULLVALUE, but also supports text fields. Salesforce will continue to support NULLVALUE, so you do not need to change existing formulas. |
| <b>PRIORVALUE</b> | Returns the previous value of a field.   |

## Logical Functions

| Function         | Description  |
|------------------|--|
| <b>AND</b>       | Returns a TRUE response if all values are true; returns a FALSE response if one or more values are false.  |
| <b>CASE</b>      | Checks a given expression against a series of values. If the expression is equal to a value, returns the corresponding result. If it is not equal to any values, it returns the <code>else_result</code> . |
| <b>IF</b>        | Determines if expressions are true or false. Returns a given value if true and another value if false.   |
| <b>ISCHANGED</b> | Compares the value of a field to the previous value and returns TRUE if the values are different. If the values are the same, this function returns FALSE.   |

| Function        | Description  |
|-----------------|--|
| <b>ISNEW</b>    | Checks if the formula is running during the creation of a new record and returns TRUE if it is. If an existing record is being updated, this function returns FALSE. |
| <b>ISNUMBER</b> | Determines if a text value is a number and returns TRUE if it is. Otherwise, it returns FALSE.   |
| <b>NOT</b>      | Returns FALSE for TRUE and TRUE for FALSE.   |
| <b>OR</b>       | Determines if expressions are true or false. Returns TRUE if any expression is true. Returns FALSE if all expressions are false.                                     |

## Math Functions

| Function           | Description  |
|--------------------|--|
| <b>ABS</b>         | Calculates the absolute value of a number. The absolute value of a number is the number without its positive or negative sign. |
| <b>CEILING</b>     | Rounds a number up to the nearest integer.   |
| <b>DISTANCE</b>    | Calculates the distance between two locations in miles or kilometers.  |
| <b>EXP</b>         | Returns a value for e raised to the power of a number you specify.   |
| <b>FLOOR</b>       | Returns a number rounded down to the nearest integer.  |
| <b>GEOLOCATION</b> | Returns a geolocation based on the provided latitude and longitude. Must be used with the DISTANCE function.                   |
| <b>LN</b>          | Returns the natural logarithm of a specified number. Natural logarithms are based on the constant e value of 2.71828182845904. |
| <b>LOG</b>         | Returns the base 10 logarithm of a number.   |
| <b>MAX</b>         | Returns the highest number from a list of numbers.   |
| <b>MIN</b>         | Returns the lowest number from a list of numbers.  |
| <b>MOD</b>         | Returns a remainder after a number is divided by a specified divisor.  |
| <b>ROUND</b>       | Returns the nearest number to a number you specify, constraining the new number by a specified number of digits.               |
| <b>SQRT</b>        | Returns the positive square root of a given number.  |

## Text Functions

| Function          | Description  |
|-------------------|--|
| <b>BEGINS</b>     | Determines if text begins with specific characters and returns TRUE if it does. Returns FALSE if it does not.              |
| <b>BR</b>         | Inserts a line break in a string of text.  |
| <b>CASESAFEID</b> | Converts a 15-character ID to a case-insensitive 18-character ID.  |
| <b>CONTAINS</b>   | Compares two arguments of text and returns TRUE if the first argument contains the second argument. If not, returns FALSE. |
| <b>FIND</b>       | Returns the position of a string within a string of text represented as a number.  |

| Function            | Description  |
|---------------------|--|
| <b>GETSESSIONID</b> | Returns the user's session ID.   |
| <b>HYPERLINK</b>    | Creates a link to a URL specified that is linkable from the text specified.  |
| <b>IMAGE</b>        | Inserts an image with alternate text and height/width specifications.  |
| <b>INCLUDES</b>     | Determines if any value selected in a multi-select picklist field equals a text literal you specify.   |
| <b>ISPICKVAL</b>    | Determines if the value of a picklist field is equal to a text literal you specify.  |
| <b>LEFT</b>         | Returns the specified number of characters from the beginning of a text string.  |
| <b>LEN</b>          | Returns the number of characters in a specified text string.   |
| <b>LOWER</b>        | Converts all letters in the specified text string to lowercase. Any characters that are not letters are unaffected by this function. Locale rules are applied if a locale is provided.                   |
| <b>LPAD</b>         | Inserts characters you specify to the left-side of a text string.  |
| <b>MID</b>          | Returns the specified number of characters from the middle of a text string given the starting position.   |
| <b>RIGHT</b>        | Returns the specified number of characters from the end of a text string.  |
| <b>RPAD</b>         | Inserts characters that you specify to the right-side of a text string.  |
| <b>SUBSTITUTE</b>   | Substitutes new text for old text in a text string.  |
| <b>TEXT</b>         | Converts a percent, number, date, date/time, or currency type field into text anywhere formulas are used. Also, converts picklist values to text in validation rules, formula fields, and field updates. |
| <b>TRIM</b>         | Removes the spaces and tabs from the beginning and end of a text string.   |
| <b>UPPER</b>        | Converts all letters in the specified text string to uppercase. Any characters that are not letters are unaffected by this function. Locale rules are applied if a locale is provided.                   |
| <b>VALUE</b>        | Converts a text string to a number.  |

## Summary Functions

The following functions are available with summary, matrix, and joined reports.

| Function              | Description  |
|-----------------------|--|
| <b>PARENTGROUPVAL</b> | This function returns the value of a specified parent grouping. A “parent” grouping is any level above the one containing the formula. You can only use this function in custom summary formulas for reports.  |
| <b>PREVGROUPVAL</b>   | This function returns the value of a specified previous grouping. A “previous” grouping is one that comes before the current grouping in the report. Choose the grouping level and increment. The increment is the number of columns or rows before the current summary. The default is 1; the maximum is 12. You can only use this function in custom summary formulas for reports. |

## Advanced Functions

| Function            | Description  |
|---------------------|--|
| <b>GETRECORDIDS</b> | Returns an array of strings in the form of record IDs for the selected records in a list, such as a list view or related list.   |
| <b>INCLUDE</b>      | Returns content from an s-control snippet. Use this function to reuse common code in many s-controls.  |
| <b>LINKTO</b>       | Returns a relative URL in the form of a link (href and anchor tags) for a custom s-control or Salesforce page.   |
| <b>REGEX</b>        | Compares a text field to a regular expression and returns TRUE if there is a match. Otherwise, it returns FALSE. A regular expression is a string used to describe a format of a string according to certain syntax rules. |
| <b>REQUIRESRIPT</b> | Returns a script tag with source for a URL you specify. Use this function when referencing the Force.com AJAX Toolkit or other JavaScript toolkits.  |
| <b>URLFOR</b>       | Returns a relative URL for an action, s-control, Visualforce page, or a file in a static resource archive in a Visualforce page.   |
| <b>VLOOKUP</b>      | Returns a value by looking up a related value on a custom object similar to the VLOOKUP() Excel function.  |

## Encoding Functions

| Function            | Description   |
|---------------------|---|
| <b>HTMLENCODE</b>   | Encodes text and merge field values for use in HTML by replacing characters that are reserved in HTML, such as the greater-than sign (>), with HTML entity equivalents, such as &gt;.   |
| <b>JSENCODE</b>     | Encodes text and merge field values for use in JavaScript by inserting escape characters, such as a backslash (\), before unsafe JavaScript characters, such as the apostrophe (').   |
| <b>JSHTMLENCODE</b> | Encodes text and merge field values for use in JavaScript within HTML tags by inserting escape characters before unsafe JavaScript characters and replacing characters that are reserved in HTML with HTML entity equivalents.  |
| <b>URLENCODE</b>    | Encodes text and merge field values for use in URLs by replacing characters that are illegal in URLs, such as blank spaces, with the code that represent those characters as defined in <i>RFC 3986, Uniform Resource Identifier (URI): Generic Syntax</i> . For example, blank spaces are replaced with %20, and exclamation points are replaced with %21. |